



**SVI**  
SPECTRAVIDEO

# C.U.C. ~ info

SPECTRAVIDEO COMPUTER USERS CLUB NL

## SV-COMPUTER GEBRUIKERS MAGAZINE

Nederland f3, = België Bfr 45 Duitsland DM3, = Suriname s/f3, =



**MICRO  
COMPUTER**

**BASIC**

- MSX
- softwarenieuws
- programma's
- lezerspost
- tricks, tips, hints
- clubnieuws
- boekennieuws
- info artikelen
- techniek
- markt

OKT/NOV

NR.

**4**

## VOORBEELDEN

```

10 KEY 1, " beep" + CHR$(13)
20 KEY 2, " list" + CHR$(13)
30 KEY 3, "screen 1"
40 KEY LIST

```

```

100 ON KEY GOSUB 300, 400, 500, 600, 700
110 :
120 FOR T = 1 TO 5: READ A#
130 KEY T, " " + A#
140 NEXT
150 :
160 DATA een, twee, drie, vier, vijf
170 :
180 KEY ON
190 :
200 A = A + 1: PRINT A: GOTO 200
210 ' -----
300 T = 1: GOTO 810
400 T = 2: GOTO 810
500 T = 3: GOTO 810
600 T = 4: GOTO 810
700 T = 5
800 :
810 IF INKEY# > "" GOTO 810
820 :
830 PRINT "Functietoets nummer:"; T
840 :
850 RETURN

```

```

10 ON KEY GOSUB 80: KEY (1) ON
20 KEY 1, ""
30 KEY STOP: PRINT "Druk F1 ";
40 FOR X = 1 TO 2500: NEXT
50 KEY ON
60 GOTO 60
70 ' -----
80 PRINT: PRINT "Programma onderbreking ";
90 RETURN

```

---

**De functietoetsen**

---

**KEY nummer, stringexpressie**

De eerste 15 karakters van de stringexpressie worden 'onder' de functietoets met het gegeven nummer (1-10) geplaatst. Van iedere toets kunnen de eerste zeven karakters onderin het beeld weergegeven worden (SCREEN 0,1).

**KEY LIST**

De momentele 'inhoud' van alle functietoetsen wordt op het beeldscherm weergegeven.

**ON KEY GOSUB lijst met regelnummers**

Deze instructie (die slechts een keer in het programma hoeft voor te komen) bepaalt de eerste regel van subroutines die aangeroepen zullen worden wanneer de betreffende functietoets wordt ingedrukt; het lopende programma wordt dan onderbroken (interrupt). De lengte van de 'lijst met regelnummers' mag variëren van 1-10 getallen, gescheiden door komma's.

**KEY (nummer) ON**

De functietoets 'interrupt' dient geactiveerd te worden met KEY ON (voor alle toetsen tegelijk) of KEY (nummer) ON (als slechts een toets geactiveerd wordt).

**KEY (nummer) OFF**

De functietoets 'interrupt' voor een of alle functietoetsen wordt uitgeschakeld.

**KEY (nummer) STOP**

De functietoets 'interrupt' voor een of alle functietoetsen wordt tijdelijk uitgeschakeld. Het indrukken van een functietoets leidt dus niet tot een programma onderbreking, maar wordt wel onthouden, zodat na de volgende KEY (nummer) ON instructie deze onderbreking alsnog plaatsvindt.

---

C.U.C.-info is het kontakt-  
orgaan van de SpectraVideo  
Computer Users Club nl.

Redactie: Wouter Alexander  
Peter Zevenhoven

Basicode: Rini Kikkert

C.U.C.-Consel voor Oost-Ned.  
Henk Hoff  
Westeinde 122  
7671 CE Vriezenveen

Lezerspost, programma's, do-  
cumentatie, opgave adver-  
tenties, copy, manuscripten,  
monsters, opgaven leden, do-  
nateurs en abonnementen:

SpectraVideo C.U.C.  
Postbus 202  
2300 AE Leiden

Betaling lidmaatschapsgeld,  
donaties, advertenties, do-  
cumentatie en lezers-service  
artikelen:

Ned. Middenstandsbank  
LEIDEN

t.g.v. SV Comp Users Club  
rek. 67.86.10.231  
giro bank: 47139

Copyrights 1984 C.U.C. nl

Zonder schriftelijke toestem-  
ming mag niets uit deze uit-  
gave worden overgenomen of  
vermenigvuldigd.

CONSUMENTENLIJN

SpectraVideo Int.

020 - 139.960

vrijdags 10.00 - 11.00 uur

Dhr. van der Roest

ARTIKELEN

- 7 Clubverslag
- 24 BASICODE corner

PRAKTIJK

- 6 Lang leve
- 9 Pythagoras
- 14 TV-tennis
- 17 Pendule
- 20 INPUT op SCREEN 1
- 21 Worteltrekken
- 23 Diamant

DIVERSEN

- 1 Colofon
- 1 Inhoud
- 3 Redactorial
- 5 I/O ports
- 16 Lezers-service
- 8 Agenda

RUBRIEKEN

- 4 Lezers in de pen
- 19 Informatie bulletin 5
- 22 Sintaks error

EXTRA

- BASIC informatie-kaart 3



# REDAKTORIAL



LS,

Het gaat goed met C.U.C. !  
De aantallen komen in onverwacht-  
te getalen binnen, de C.U.C.-  
computerdagen verheugen zich in  
een zeer grote belangstelling, de  
correspondentie is huizehoog en  
de lezers-service heeft zichzelf  
al meer dan bewezen.

Een schriftelijke enquette op  
een van de computerdagen heeft  
aangetoond, dat de meesten zich  
interesseren voor de SVI computer  
in zijn totaliteit; een minderheid  
heeft interesse in een deelgebied.  
De inhoud van ons info zal daar  
natuurlijk op worden afgestemd.

MSX is een hoofdstuk apart; maar  
dat wisten we uiteraard al. Hoe  
uw redactie dit deelgebied in het  
computerwereldje meent te moeten  
aanpakken, zal in deze uitgave al  
wat meer gestalte krijgen.

Op het moment leest u al weer het  
4e nummer, waarin als nieuwigheid  
een stukje hardware.

Tot besluit: blijf ideeën, denk-  
beelden, suggesties en software  
insturen - en bezoek de computer-  
dagen. Des te meer genieten we  
allemaal van onze hobby !

Beste Spectravideoïsten,

Daar zitten we, "oude" SV.328 gebruikers, nu zo na de Firato. Een stuk wijzer en een illusie armer. De SV.328 is, zoals we reeds lang vreesden, GEEN MSX computer. Voor sommigen geen probleem, maar voor anderen, waaronder ikzelf, WEL.

In dit blad lijkt mij tusser al het lof, ook een woord van kritiek op zijn plaats. De Firato: het is nu officieel geworden, de SV.328 is geen MSX computer. Daar zit je nu met je de bodem ingeslagen verwachtingen.

Beste C.U.C.-redactie,

Misschien kan deze brief worden geplaatst in het clubinfo. Aangezien er wel meer zullen zijn die met de zelfde problematiek zitten. Het feit dat er meerdere zijn en dat de ontwikkelingen gevolgd gaan worden helpt in ieder geval een beetje.

Zet het goede werk voort en de groetjes van

Hans v d Meer  
Vechtstr 78 hs  
1079 JL AMSTERDAM

4 Red.:

Een clubblad, zoals het onze, is een soort podium. Als je er in komt te staan, ziet iedereen je en ook hoort ie-

dereden je, of wat je te vertellen hebt terecht is of niet. Iedere toehoorder zal op zijn eigen wijze reageren en zo ook de redactie. De brief van Hans, zoals afgedrukt, is een uittreksel. Misschien zijn er meer die e.e.a. zo ervaren, maar of dat op feitelijke gronden berust? Verderop is er meer over te lezen.

Geachte Heer,

Hartelijk dank voor de C.U.C.-info's. Ik ben erg tevreden met dit clubblad. Jammer dat men telkens 2 maanden moet wachten. Ik vindt het fantastisch dat er een SVI-club is. Er zijn hier twee boeken in het Duits over de SVI computers verschenen. Verder hoop ik dat de club flink mag groeien.

L. Cuyvers  
Stressemanstr 14  
4152 KEMPEN Duitsland

Red.:

Wij horen veel positieve reacties op onze activiteiten. Nu zelfs uit Duitland. Bij deze vermelden wij tevens een abonnee in Israël in het bestand te hebben. Wie gaat er nog verder wonen?

I/O	PORT	R/W	DESCRIPTION	REMARKS
10H		W	WRITE DATA PORT	PRINTER
11H		W	DATA STROBE	PRINTER
12H		R	STATUS (BIT 0 = "0" FOR READY)	PRINTER
20H		R	RECEIVER BUFFER REGISTER	MODEM
		W	DIVISOR LATCH (LEAST SIGNIFIC.)	MODEM
		W	TRANSMITTER HOLDING BUFFER REG.	MODEM
21H		W	DIVISOR LATCH (MOST SIGNIFICANT)	MODEM
		W	INTERRUPT ENABLE REGISTER	MODEM
22H		W	INTERRUPT IDENTIFICATION REG.	MODEM
23H		W	LINE CONTROL REGISTER	MODEM
24H		W	READ MODEM CONTROL REGISTER	MODEM
25H		R	LINE STATUS REGISTER	MODEM
26H		R	READ MODEM STATUS REGISTER	MODEM
28H		R	RECEIVER BUFFER REG (LEAST SIG.)	RS-232
		W	DIVISOR BUFFER REGISTER	RS-232
		W	TRANSMITTER HOLDING BUFFER REG.	RS-232
29H		W	DIVISOR LATCH (MOST SIGNIFICANT)	RS-232
		W	INTERRUPT ENABLE REGISTER	RS-232
2AH		W	INTERRUPT IDENTIFICATION REG.	RS-232
2BH		W	LINE CONTROL REGISTER	RS-232
2CH		W	MODEM CONTROL REGISTER	RS-232
2DH		R	LINE STATUS REGISTER	RS-232
2EH		R	MODEM STATUS REGISTER	RS-232
30H		R	FD-1793 STATUS REGISTER	FLOPPY DISK
		W	FD-1793 COMMAND REGISTER	FLOPPY DISK
31H		R/W	FD-1793 TRACK REGISTER	FLOPPY DISK
32H		R/W	FD-1793 SECTOR REGISTER	FLOPPY DISK
33H		R/W	FD-1793 DATA REGISTER	FLOPPY DISK
34H		R	READ INTRO AND DRQ O/P PINS	FLOPPY DISK
		W	DISK SELECT REGISTER (BIT 0 = "0" TO SELECT DISK 1 BIT 1 = "0" TO SELECT DISK 2)	FLOPPY DISK
38H		W	DENSITY SELECT REGISTER (BIT 0 = "0" FOR DOUBLE DEN. BIT 0 = "1" FOR SINGLE DEN.)	FLOPPY DISK
50H		W	ADDRESS REGISTER SELECT	80-COLUMN CARD
51H		W	CRT CONTROLLER REGISTER (R0-R17)	80-COLUMN CARD
58H		W	CRT BANK CONTROL ( = OFFH BANK ON, = OOH BANK OFF)	80-COLUMN CARD
80H		W	TMS-9918A WRITE MODE=0	VDP
81H		W	TMS-9918A WRITE MODE=1	VDP
84H		R	TMS-9918A READ MODE=0	VDP
85H		R	TMS-9918A READ MODE=1	VDP
88H		W	AY-3-8910 LATCH ADDRESS	PSG
8CH		W	AY-3-8910 WRITE	PSG
90H		R	AY-3-8910 READ	PSG
96H		W	WRITE 8255 PORT C	PPI
97H		W	WRITE 8255 CONTROL WORD REG.	PPI
98H		R	READ 8255 PORT A	PPI
99H		R	READ 8255 PORT B	PPI

```

100 PRINT "Lang zal ze leven."
110 PRINT
120 PRINT"Ze leve hoog."
130 PRINT
140 READ A$
150 READ B$
160 READ C$
170 PLAY A$
180 PLAY B$
190 PLAY C$
200 DATA s11 r1 m80 o5 14c 18 cc 14 c
      o4g o5e 18ee 14 ecg 18 ggagfe
      14 ddd 18 gf 12 efg 14 af 12
      edc.
210 DATA r2
220 DATA s11 r2 m90 o4 18d 18d 18d 14
      ggga 12 b 18 gga 14 bbb o5c
      12 d r8 14e 12d r8 14 e 12d r8
      14e 12d f+ m112 g...
230 END

```

**c.d.systems** computer  
vachandel

berkhoutlaan 2c 2161em lisse 02521-17459

SOFTWARE MEMORY EXPENSION	f 35,--
TAKISTO SCOOP	35,--
ZEESLAG	35,--
ESVAY	35,--
WOORDENSPEL	35,--
TEKEN-programma	35,--
CP/M pakket 160 Kbyte	999,--
Ombouw SVI.318 naar 64k	325,--
Zeer binnenkort: SVI-netwerk	
Prof. boekhoudpakket	
2-printer switch	

**berkhoutlaan 2c 2161em lisse 02521-17459**



Nu we reeds twee C.U.C.-dagen achter de rug hebben, willen we graag onze indruk meegeven aan de leden die (nog) niet in de gelegenheid waren een van onze computerdagen bij te wonen.

Wij - de redactie - hebben deze bijeenkomsten als zeer, zeer positief ervaren: we krijgen meer vragen dan we kunnen beantwoorden en dat bewijst dat de club leeft. Regelmatig ontstaan er discussie's over zeer uiteenlopende onderwerpen met als centraal thema onze "SpectraVideo's".

Het succes (en de grote opkomst) is te danken aan diverse zaken:

1e EEN ZEER GOEDE OPVANG bij C.D. SYSTEMS te LISSE. Dit zetten we bewust bovenaan, want de accommodatie is uitstekend en de koffie prima.

2e EEN COMPUTER met ONGEKENDE MOGELIJKHEDEN. Op de bijeenkomst van 6 oktober j.l. was b.v. te zien hoe men telex, morse enz. kon ontvangen d.m.v. een uitbreiding via de joystick-ingang; meer hierover in te toekomst.

3e EEN POSITIEVE HOUDING van de AANWEZIGEN. Zoveel mensen met zoveel meningen geeft wel eens heftige discussies, maar het uitgangspunt is voor eenieder gelijk; men wil zo efficiënt mogelijk met z'n computer omgaan, hoe de configuratie ook moge zijn.

Op punt 3 willen we terugkomen en een punt van discussie hier melden, te weten het feit, dat de SV.318/.328(MKII) niet de standaard MSX-computer is (zoals gesuggereerd in de folders?).

Hoewel het op de eerste plaats niet onze taak is hieraan iets te

doen (wel des dealers, via de importeur en de importeur weer bij de fabrikant) kunt u er van overtuigd zijn, dat wij al het mogelijke in 't werk zullen stellen om de SV.318/.328(MKII) ook als standaard MSX-computer te kunnen laten werken:

maar we houden u op de hoogte. We denken wel, dat de SV.318/.328 veel meer kan dan de standaard-MSX'ers!!

(Laatste nieuws: middels een op korte termijn te verkrijgen adapter, zullen de .318 en .328 machines volledig MSX compatibel worden (hetgeen in de folders is gesuggereerd). Het zal tevens mogelijk zijn daardoor een standaard cassette recorder aan te sluiten. Wij verwachten tevens, dat te zijner tijd een tweede adapter zal worden aangeboden die de uitbreidbaarheid van de semi-professionele SpectraVideo machines handhaaft).

NU NOG EVEN IETS HEEL BELANGRIJKS. De C.U.C.-computerdagen werden (en worden) nu gehouden in Lisse en wellicht later ook elders in de Randstad (zie de AGENDA). Maar wij beseffen heel goed, dat dit voor de meeste leden te ver is of moeilijk bereikenbaar.

Bij deze doen wij dan ook een beroep op de SpectraVideo-dealers elders in het land om (zaterdag?) ruimte ter beschikking te stellen voor C.U.C.-dagen t.b.v. hun regio's. Het dagelijks bestuur zal zeker een eerste computerdag bij bijv. een SVI-dealer helpen organiseren en bijwonen. Tevens doen wij een beroep op de leden in die regio's zich beschikbaar te stellen om de zaak aldaar te coördineren.

LAAT HET ZO SPOEDIG MOGELIJK WETEN VIA ONZE POSTBUS. (twee nul twee - leiden).

Deze beroemde Griekse filosoof en wiskundige heeft in ieder geval een meetkundige stelling op zijn naam gebracht.

Het beginsel er van was in feite al aan de Babyoniërs bekend, maar heeft via de Griek zich nu ook uitgebreid tot een groot deel van de moderne wereldbevolking.

Ook in computerland is de stelling vaak besproken en beprogrammeerd, maar, naar ons weten, nog niet op de (educatieve) wijze zoals dit programma u uit de bytes doet.

Op scherm 1 zien we hoe de stelling van Pyt zich ontwikkelt van het opgeven van de zijden tot en met het aanschouwelijk voorstellen van hoe daarna e.e.a. in zijn werk gaat. Interessant voor de computeraars en een mooi huishoudelijk hulpje voor de wiskunde leraren.

Wie ontwikkelt nu de kegelsneden?




```

1000 REM De stelling van Pythagoras
1010 :
1020 REM Na CTRL/STOP ....
1030 ON STOP GOSUB 6000: STOP ON
1040 :
1050 REM In geval van fouten ....
1060 ON ERROR GOTO 7000
1070 :
1080 COLOR 1,12,12: SCREEN 1: DEFSNG A-Z
1090 :
1100 REM Deel scherm in in 4 vlakken
1110 FOR X=0 TO 3: GOSUB 5030: NEXT
1120 :
1130 REM Tekening van kwadraat teken
1140 TW$="E1 F1 D1 G2 D1 R2
1150 :
1160 REM teken driehoek
1170 DRAW "C1 S0 BM 170,10 D50 R50 M 170,10
1180 :
1190 LOCATE 150,32: PRINT "A
1200 LOCATE 200,20: PRINT "C
1210 LOCATE 192,70: PRINT "B
1220 :
1230 LOCATE 20,20: PRINT "Stelling van
1240 LOCATE 20,29: PRINT "Pythagoras:
1250 LOCATE 20,46: PRINT "A + B = C
1260 '
1270 REM Vul kwadraat tekens in
1280 FOR T= 27 TO 87 STEP 30
1290 DRAW "BM =T; , 44 X TW$;
1300 NEXT
1310 ' -----
1320 LOCATE 20,110: PRINT "Welke zyden
1330 LOCATE 20,120: PRINT "zijn bekend?
1340 LOCATE 8,130: PRINT "Toets
1350 LOCATE 20,140: PRINT "1: A en B
1360 LOCATE 20,150: PRINT "2: A en C
1370 LOCATE 20,160: PRINT "3: B en C
1380 :
1390 REM Wacht op opdracht van gebruiker
1400 :
1410 I=VAL(INKEY$):IF I<1 OR I>3 GOTO 1410
1420 ' -----
1430 REM Pas formule aan gewenste berekening aan
1440 R$=CHR$(68-I): B$="+
1450 L1$="A": IF I>1 THEN L1$="C": B$="--
1460 L2$="B": IF I=2 THEN L2$="A"
1470 :
1480 LOCATE 32,60: PRINT "Afleiding:"
1490 LOCATE 20,80: PRINT R$; " = ";
1500 DRAW "BM 38,80 M 40,88 M 43,74 R50 F2
1510 PRINT L1$; " ";B$; " ";L2$
1520 :
1530 REM Vul kwadraat tekens in
1540 DRAW "BM 51,78 X TW$; BM 81,78 X TW$;"
1550 ' -----
1560 REM Vraag lengte van zyden op
1570 :
1580 X=2: GOSUB 5030:' Wis venster
    
```

```

1590 LOCATE 20,110:PRINT "Lengten (cm)?
1600 LOCATE 20,125: PRINT "Zyde: ";L1%;"= ";
1610 GOSUB 5070: L1=VAL(IN%)
1620 LOCATE 20,135: PRINT "Zyde: ";L2%;"= ";
1630 GOSUB 5070: L2=VAL(IN%)
1640 :
1650 REM Controleer waarden op nul
1660 IF L2*L1=0 THEN ERRDR 11
1670 ' -----
1680 REM Bereken de lengte van de derde zyde
1690 ON I GOTO 1700, 1710, 1720
1700 LA=L1: LB=L2: Q=LA^2+LB^2: LC=SQR(Q): LR=LC: GOTO 1740
1710 LC=L1: LA=L2: Q=LC^2-LA^2: LB=SQR(Q): LR=LB: GOTO 1740
1720 LC=L1: LB=L2: Q=LC^2-LB^2: LA=SQR(Q): LR=LA
1730 ' -----
1740 REM Bereken de schaal
1750 S=30/LA: IF LALB THEN S=30/LB
1760 LA=INT(S*LA): LB=INT(S*LB): LC=INT(S*LC)
1770 :
1780 REM Teken de driehoek met vierkante vlakken
1790 LINE (167,160)-STEP(-LA,-LA), 14, BF
1800 LINE (167,160)-STEP( LB, LB), 10, BF
1810 B=167+SB
1820 LINE (167+LB,159) -STEP(-LB,-LA), 7
1830 LINE -STEP( LA,-LB), 7
1840 LINE -STEP( LB, LA), 7
1850 LINE -STEP(-LA, LB), 7
1860 PAINT STEP(-LB/2+1,-LA/2-1), 7
1870 ' -----
1880 REM Simuleer de berekening visueel
1890 X=1: GOSUB 5030: LOCATE 134,4
1900 COLOR 15: PRINT "Het oppervlak van ";
1910 IF I > 1 GOTO 2070
1920 :
1930 REM A^2 + B^2
1940 PRINT "A": LOCATE 140,14
1950 LINE (136,34) -STEP(LA,LA), 14, BF
1960 GOSUB 5340: PRINT "plus dat van B
1970 LINE (190,34) -STEP(LB,LB), 10, BF
1980 :
1990 REM Verklein vlak B en vergroot A
2000 FOR T=LB+1 TO 1 STEP -1
2010 LINE (189,33) -STEP( T, T), 6, B
2020 H=(LC-LA)/LB*(LB+1-T)+LA
2030 LINE (136,34) -STEP( H, H), 14, B
2040 FOR H=0 TO 99: NEXT
2050 NEXT T: GOTO 2210
2060 :
2070 REM C^2 - (A^2 of B^2)
2080 PRINT "C": LOCATE 140,14
2090 LINE (136,34) -STEP(LC,LC), 7, BF
2100 GOSUB 5340: PRINT "min dat van "; L2%
2110 :
2120 REM Verklein vlak C en vergroot A of B
2130 LT=LA: L3=LB: IF I=3 THEN SWAP LT,L3
2140 FOR T=1 TO LT+1
2150 LINE (136,92)- STEP( T,-T), 22-4*I, B
2160 H=(LC-L3)/LT*(LT+1-T)+L3+2
2170 LINE (135,33)-STEP( H, H),6,B
2180 FOR H=0 TO 99: NEXT
2190 NEXT T
2200 :
2210 REM Teken oppervlak van berekende zijde

```



```

2220 GOSUB 5340: GOSUB 5390: LOCATE 134,4
2230 PRINT "Is gelijk aan het":LOCATE 140,14
2240 PRINT "oppervlak van "; R$
2250 LINE (190,34)-STEP (S*LR,S*LR),7+3.5*(I-1),BF
2260 ' -----
2270 REM Conclusie
2280 GOSUB 5340: GOSUB 5340
2290 U$= "Vlak !: ####.## $ ####.## = #####.##
2300 COLOR 11: LINE (0,96) - (255,191),12,BF
2310 LOCATE 110,100: PRINT "RESULTAAT:
2320 LOCATE 14,112: PRINT USING U$; L1$, L1, L1, L1*L1
2330 LOCATE 230,120: PRINT B$
2340 LOCATE 14,128: PRINT USING U$; L2$, L2, L2, L2*L2
2350 LOCATE 182,135: PRINT "-----
2360 LOCATE 74,142: PRINT "Oppervlak van ";R$;" = ";
2370 PRINT USING "#####.##"; Q
2380 COLOR 15: R=SGR(Q)
2390 LOCATE 20,160: PRINT "De wortel van"; Q; "is"; R
2400 LOCATE 20,170: PRINT "De lengte van zyde ";R$;
2410 PRINT " is dus";R;"cm.
2420 ' -----
2430 REM Nog eens?
2440 COLOR 1: LOCATE 60,184
2450 PRINT "Nog een berekening? J/N
2460 ON 1+INSTR("-JjNn",INKEY$)\2 GOTO 2460, 1100, 6000
5000 ' -----
5010 ' SUBROUTINES
5020 ' -----
5030 'Wis schermgedeelte
5040 LINE ((X AND 1)*128,(X AND 2)*48)-STEP(127,96),5+X,BF
5050 RETURN
5060 ' -----
5070 REM LINE INPUT IN$ op scherm 1
5080 B=74: IN$="": L=0: C=1
5090 :
5100 ' Wacht op toets, laat cursor knipperen
5110 COLOR C: LOCATE B: PRINT CHR$(201);; LOCATE B
5120 IF TIME > 15 THEN TIME = 0: C = C XOR 6
5130 I$=INKEY$: IF I$="" GOTO 5110
5140 :
5150 ' Cursor uit
5160 COLOR 7: PRINT CHR$(201);; COLOR 1: LOCATE B
5170 :
5180 ' Is toets enter, backspace, cijfer of punt?
5190 IF I$=CHR$(13) THEN RETURN
5200 IF I$<>CHR$(8) OR L=0 GOTO 5250
5210 :
5220 REM Backspace : wis laatste karakter
5230 L=L-1: IN$=LEFT$(IN$,L): B=B-6: GOTO 5110
5240 :
5250 IF L=4 OR I$>"9" GOTO 5310
5260 IF I$<"0" AND I$<>".," GOTO 5310
5270 :
5280 REM Punt of cijfer : voeg karakter aan IN$ toe
5290 PRINT I$;: L=L+1: IN$=IN$+I$: B=B+6: GOTO 5110
5300 :
5310 REM Verboden karakter of te lang cijfer: beep
5320 BEEP: GOTO 5110
5330 ' -----
5340 REM Wacht ca. 2 seconden
5350 TIME = 0
5360 IF TIME < 100 GOTO 5360
5370 RETURN

```

$A^2 - B^2 = C^2$

```

5380 ' -----
5390 REM Wis tekstvenster rechtsboven
5400 LINE (129,1)-(255,25), 6, BF
5410 RETURN
6000 ' -----
6010 ' CTRL/STOP regel
6020 ' -----
6030 REM Herstel standaard kleuren
6040 COLOR 15,4,5
6050 END
7000 ' -----
7010 ' FOUTAFHANDELING
7020 ' -----
7030 IF ERR<>11 OR ERL<>1660 GOTO 7080
7040 :
7050 REM Als voor een zyde nul gegeven is.
7060 ER$="Waarde kan geen nul zijn!": GOTO 7130
7070 :
7080 IF ERR<>5 OR ERL<>1710 AND ERL<>1720 GOTO 7190
7090 :
7100 REM Als voor C een te lage waarde gegeven is.
7110 ER$="C moet hoger zijn dan "+CHR$(ASC(R$) XOR 3)
7120 :
7130 REM Geef foutmelding en vraag waarden opnieuw
7140 LINE (20,180) - (230,191),10,BF
7150 LOCATE 125-3*LEN(ER$),183: PRINT ER$
7160 GOSUB 5340: X=3: GOSUB 5030
7170 RESUME 1560
7180 :
7190 REM Als er geen oplossing is voor een fout...
7200 ON ERROR GOTO 0:' Standaard foutmelding
7210 ' -----
7220 ' P.Z
7230 END

```

Kom ook eens naar de SpectraVideo-dag  
 ALTIJD DE LAATSTE NIEUWTJES  
 en ALTIJD actueel met software,  
 hardware en boeken

M.I.V. 1 Sept. organiseren wij maandelijks op de eerste zaterdag van de maand:  
 DE SPECTRA VIDEO DAG  
 Apparatuur, programmatuur en boeken worden op bovenstaand adres zo volledig  
 mogelijk gepresenteerd.

## DE COMPUTER WINKEL VOOR DE BOLLENSTREEK

berkhoutlaan 2c 2161em lisse 02521-17459

Hebt U de gereserveerde woordenlijst (C.U.C. info nr. 1, pagina 5) gezien? Een van de vele kreten op die pagina is "PDL".

Dat staat voor PADDLE, dacht ik, en direct werden 2 potmeters, draden en pluggen (voor de joystickpoorten) aangeschaft. Helaas, wat ik ook deed, de computer weigerde om ook maar iets te presteren wat op het inlezen van een PADDLE leek.


Een van de allereerste programma's die ik voor mijn SV.328 schreef was een Z-80 disassembler. Daarmee heb ik enkele dagen in het BASIC-ROM gePEEKt tot ik de PDL routine vond.

Wat blijkt nu? Na het woord PDL moet, tussen haakjes, een getal (1-4) gegeven worden. Dit getal wijst een EOC of SENSE ingang van joystickpoort 1 of 2 aan [zie appendix E van de handleiding (dan is die toch nog ergens goed voor)].

Na de aanroep van PDL worden, gedurende 6 microseconden, de FORWARD/SCK uitgangen van beide joystickpoorten 0 Volt gemaakt (deze zijn normaal 5 Volt). Hierna wacht de PDL routine tot opgegeven ingang (het cijfer na PDL) 0 Volt wordt, terwijl intussen een teller de stand bijhoudt.

Conclusie: aan de joystickpoort dient een stukje electronica te komen, waarna, met de PDL instructie, men de momentele stand van een potentiometer kan opvragen (een getal van 0 tot 255). Heel mooi dus.


Bekijk het volgende schema eens:



```

100 REM "TV tennis"
110 COLOR 14, 1, 1: SCREEN 2, 2: SOUND 1, 0
120 DEFINT A-Z: S = RND(-TIME) + .5
130 REM Racket
140 SPRITE$(0) = STRING$(16, 15)
150 PUT SPRITE 0, , 15, 0
160 PUT SPRITE 1, , 15, 0
170 REM Bal
180 SPRITE$(1) = CHR$(6) + STRING$(3, 15) + CHR$(6)
190 REM Teken het veld
200 LINE (0, 0) - (255, 0): LINE (0, 191) - (255, 191)
210 FOR T = 0 TO 9
220 LINE (124, 20 * T) - (124, 20 * T + 8)
230 NEXT
240 REM Score bijwerken
250 LINE ( 40, 4) - (120, 40), 1, BF
260 COLOR 7: LOCATE 72, 8: PRINT USING "##"; S1
270 LINE (128, 4) - (220, 40), 1, BF
280 COLOR 9: LOCATE 138, 8: PRINT USING "##"; S2
290 REM Willekeurige 1e toestand van bal
300 H = S * 260 - 12: V = RND(1) * 160 + 20
310 R = 1: IF S THEN R = -1
320 E = 1: IF RND(1) > .5 THEN E = -1
330 REM Serveer vertraging
340 RV = RND(1) * 100 + 50
350 RV = RV - 1
360 REM Synchroniseer en ev. geluid uit
370 IF TIME THEN SOUND 8,0: TIME = 0 ELSE 370
380 REM Lees en plaats de rackets
390 P1 = (P1 + PDL(1)) \ 2 : PUT SPRITE 0,( 12, P1 - 20)
400 P2 = (P2 + PDL(3)) \ 2 : PUT SPRITE 1,(232, P2 - 20)
410 IF RV GOTO 350
420 REM Zijkant? Verticale richting veranderen
430 IF V < 5 OR V > 181 THEN E = -E: SOUND 0, 100: SOUND 8, 10
440 REM Bereken nieuwe balpositie
450 H = H + 4 * R: V = V + E: PUT SPRITE 2, (H, V), 15, 1
460 REM Bal tegen racket?
470 IF R * H = -16 THEN T = P1 - V: GOTO 550
480 IF R * H = 228 THEN T = P2 - V: GOTO 550
490 REM Heeft iemand een punt gemaakt? Nee, ga door
500 IF H > -10 AND H < 252 GOTO 360
510 REM Ja, punt erbij
520 PLAY "o6 v10 116 t250 c"
530 IF H > 128 THEN S2 = S2 + 1: S = 0 ELSE S1 = S1 + 1: S = 1
540 GOTO 240
550 REM Bal tegen racket? Nee, ga door
560 IF T > 23 OR T < 4 GOTO 490
570 REM Ja, verander richting
580 SOUND 0, 130: SOUND 8, 10: R = -R
590 IF T > 21 THEN E = -3: GOTO 360
600 IF T < 8 THEN E = 3: GOTO 360
610 IF T < 14 AND T > 10 THEN E = SGN(E)
620 GOTO 360
630 ' -----
640 ' P.Z.
650 END

```



Zodra op SCK een 0 Volt puls verschiijnt, wordt condensator C door transistor T opgeladen (R1 is de basisweerstand). Na het weer 5 Volt worden van SCK, zullen R2, I en P de condensator C ontladen. De spanning over C wordt via SENSE aan de computer doorgegeven, welke telt hoelang het ontlad-

den van C duurt.

De weerstand R2 is een beveiliging tegen te grote stromen door de transistor, de potmeter P is de "game paddle" en de instelpotmeter I moet even afgeregeld worden.

De knutselaars onder U zullen geen problemen hebben met het in elkaar zetten van dit schakelingetje (hoop ik).

Het afregelen van instelpotmeter I gaat als volgt:

- 1) Sluit de schakeling(en) aan op de joystickpoort(en).
- 2) Tik in:    10 LOCATE 0,0: PRINT PDL(1)  
              20 PRINT PDL(3): GOTO 10  
              RUN
- 3) Zet potmeter P op de maximale weerstand en draai aan de instelpotmeter totdat de getallen op uw scherm ongeveer 250 aangeven.

Een voorbeeld van wat met game paddles gedaan kan worden is het nu volgende programma "TV-TENNIS". Tik dat maar eens in, sluit de "home-made game paddles" aan en: VEEL PLEZIER!!

☆ ☆

#### C.U.C. lezers-service

D.01 Map testrapporten (Ned. vert.)	f 7,50
D.02 Los nr. C.U.C.-info 1	3,-
B.02 Ned. talige handleiding (eerste gedeelte)	10,-
B.03 Idem, (tweede gedeelte)	10,-
B.04 Idem, (eerste en tweede gedeelte)	18,-
C.01 BASICODE lees/schrijfprogr. op cass.	6,50
D.03 Los nr. C.U.C.-info 2/3	6,-
D.04 Los nr. C.U.C.-info 4	4,-

Levering: na vooruitbetaling + f3,- verzendkosten.  
Graag schrift. bestellen om puzzles te voorkomen.  
Zie o.a. bestelkaart achterin.



```

100 CLS: REM Wijzer klok met gong
105 '
110 REM 2 functies om de wijzer standen te berekenen
115 DEF FN SX (X,Y)= 128.5 + .75 * Y * SIN(X)
120 DEF FN CX (X,Y)= 96.5 - Y * COS(X)
125 :
130 REM Constante van 2 * PI
135 PI2 = 6.28319
140 :
145 REM Geef oude wijzerpunten een eerste waarde
150 H1 = 128: H2 = H1: H3 = H1
155 V1 = 96: V2 = V1: V3 = V1
160 '
165 INPUT "Hoe laat is het (uur,min)"; UUR, MIN
170 '
175 REM Teken de klok
180 COLOR 6,4,4: SCREEN 1: A = 1.33
185 :
190 REM De 'kast'
195 CIRCLE (128, 96), 96,, 0, 3.15, A
200 CIRCLE ( 33, 96), 30,, 4.74, 6.28, A
205 CIRCLE (223, 96), 30,, 3.14, 4.71, A
210 CIRCLE ( 35, 156), 30,, 1.57, 3.15, A
215 CIRCLE (221, 156), 30,, 0, 1.58, A
220 :
225 LINE (12, 156) - (244, 191),, BF: PAINT (128, 155)
230 :
235 REM De wijzerplaat
240 CIRCLE (128, 96), 86, 10,, A: PAINT (128, 11), 10
245 :
250 REM Cijfers op wijzerplaat
255 FOR TELLER = 1 TO 12
260 HULP = TELLER / 12 * PI2
265 HULP% = MID$(STR$(TELLER), 2)
270 LOCATE 1 + FN SX (HULP, 77) - 3 * LEN (HULP%)
275 LOCATE , FN CX (HULP, 77) - 2
280 COLOR 1 : PRINT HULP%
285 NEXT TELLER
290 '
295 REM INTERVAL houdt de tijd bij
300 DN INTERVAL = 50 GOSUB 585: INTERVAL DN
305 '
310 REM Bereken de wijzerstanden
315 HULP = SEC / 60 * PI2
320 SH = FN SX (HULP, 70)
325 SV = FN CX (HULP, 70)
330 HULP = (MIN * 60 + SEC) / 3600 * PI2
335 MH = FN SX (HULP, 63)
340 MV = FN CX (HULP, 63)
345 HULP = (UUR * 60 + MIN) / 720 * PI2
350 UH = FN SX (HULP, 50)
355 UV = FN CX (HULP, 50)
360 :
365 REM Half uur? Zorg voor gongslag
370 IF MIN = 30 AND SEC = 0 THEN SLAG = 1
375 :
380 REM Heel uur? Aantal gongslagen
385 IF MIN+SEC = 0 THEN SLAG = 12: IF UUR THEN SLAG = UUR
390 :
395 REM Als wijzerstand veranderd is, wis dan eerst
400 ' de oude wijzer. Teken daarna alle wijzers.
405 IF UH = H3 AND UV = V3 GOTO 425
410 LINE (128, 96) - (H3, V3), 10
415 H3 = UH: V3 = UV
420 :
425 IF MH = H2 AND MV = V2 GOTO 445
430 LINE (126, 96) - (H2, V2), 10
435 H2 = MH: V2 = MV
440 :
445 REM Wis de tweede wijzer altijd
450 LINE (128, 96) - (H1, V1), 10
455 H1 = SH: V1 = SV
460 :
465 REM Teken alle wijzers
470 LINE (128, 96) - (SH, SV)
475 LINE (128, 96) - (MH, MV)
480 LINE (128, 96) - (UH, UV)
485 :
490 REM Wacht tot tweede voorbij, is
495 IF SEC = 05 GOTO 495 ELSE 05 = SEC
500 :
505 REM Seconde klik
510 OUT 151, 15: OUT 151, 14
515 :
520 REM Moet de gong klinken?
525 IF SLAG = 0 OR (SEC AND 1) GOTO 310
530 :
535 REM Sla op de gong, teller - 1
540 SOUND 0, 159: SOUND 1, 0
545 SOUND 2, 123: SOUND 3, 1
550 SOUND 4, 119: SOUND 5, 0
555 SOUND 8, 16: SOUND 9, 16
560 SOUND 10, 16: SOUND 11, 0
565 SOUND 12, 99: SOUND 13, 1
570 :
575 SLAG = SLAG - 1: GOTO 310
580 '
585 REM Verhoog de tijd een seconde
590 SEC = (SEC + 1) MOD 60
595 IF SEC THEN RETURN
600 MIN = (MIN + 1) MOD 60
605 IF MIN THEN RETURN
610 UUR = (UUR + 1) MOD 12
615 RETURN
620 '
625 ' P.2.
630 END

```



- \* De volgende programma's zijn iets moeilijker dan we gewend waren.
- \* Tik ze nauwkeurig over en je moeite wordt weldra beloond.

```

100 REM woorden en getallen sorteren
110 CLS: PRINT "Tik max. 30 woorden in"
120 PRINT
130 PRINT " (a + ENT = ga sorteren.)"
140 PRINT: DIM A$(30)
150 FOR Z = 1 TO 30
160 INPUT A$(Z)
170 IF A$(Z) = "a" THEN L=Z-1:GOTO 200
180 NEXT Z
190 L = 30
200 REM bubble sort
210 FOR Z = 1 TO L - 1
220 FOR Y = Z + 1 TO L
230 IF A$(Z) > A$(Y) THEN SWAP
    A$(Z), A$(Y)
240 NEXT Y, Z
250 CLS: REM en nu sorteren
260 PRINT " Gesorteerde lijst:": PRINT
270 FOR Z = 1 TO L: PRINT A$(Z),: NEXT
280 END
    
```


\*\*

```

100 REM knippen CAPS/LOCK led
110 REM demo van een OUT-opdracht
120 WIDTH 40: SCREEN0,0: LOCATE,,0
130 :
140 ' led uit
150 A = &B11011111
160 OUT &H88, 15
170 OUT &HC, A
180 FOR M = 1 TO 48: LOCATE24,20: PRINT
    " □ □": NEXT: 'tjdus (vertraging)
190 :
200 ' led aan
210 A = &B11111111
220 OUT &H88, 15
230 OUT &HC, A
240 FOR N = 1 TO 48: LOCATE24,20: PRINT
    " ■ ■": NEXT: 'tjdus (vertraging)
250 GOTO 150
260 END
    
```

\*\*


Natuurlijk daarna RUN en ENTER geven.



```

100 REM twee zich verplaatsende cirkels
110 REM W.A. ++0484++
120 COLOR,1,9:SCREEN1 'achtergrondkleur
125 : 'en scherm kiezen
130 WIDTH 40 '40 tekens per regel
140 X=20: Y=136 'midepunt 1e cirkel
150 M=20: N=50 'midepunt 2e cirkel
160 :
170 X = X + 3 'rekenregels nieuwe-
180 M = M + 3 'midepunten
190 CIRCLE(X,Y),25,10,,,1.25
200 CIRCLE(M,N),25,14,,,1.25
210 :
220 IF X > 80 THEN Y=Y-3 'voorwaarden
230 IF M > 80 THEN N=N+3 'voor richting
240 IF X>162 THEN Y=50 'verandering
250 IF M>162 THEN N=136 'oirkelver-
260 IF X>220 THEN X=220 'plaatsing-
270 IF M>220 THEN M=220 'en.
280 :
290 GOTO 170
300 END
    
```

\*\*



samenstelling: Wouter Alexander

Wel eens geprobeerd om, nadat u een mooie tekening op SCREEN 1 getekend had, om met de INPUT of LINE INPUT instructie gegevens op te vragen? Jammer dus.

De INPUT instructie schakelt automatisch terug naar SCREEN 0 (dag tekening) en de LINE INPUT instructie maakt er helemaal een mooi rommeltje van na het drukken op ENTER of een cursortoets.

Voor degenen die, terwijl er een tekening op het scherm staat, toch gegevens willen intoetsen, is de volgende sub-routine ontworpen:

```

9000 REM LINE INPUT LI$ op SCREEN 1
9010 HOX = PEEK(&HFA04):' Cursor's horizontale positie
9020 VEX = PEEK(&HFA03):' Cursor's verticale positie
9030 VGX = PEEK(&HFA0A):' Huidige voorgrondkleur
9040 AGX = PEEK(&HFA0B):' Huidige achtergrondkleur
9050 LI$ = "" : ' Wis de string
9060 REM Plaats (print) cursor en wacht op een toets
9070 COLOR VGX: LOCATE HOX, VEX: PRINT CHR*(201)
9080 I$ = INPUT*(1): COLOR AGX
9090 REM Cursor uit
9100 LOCATE HOX, VEX: PRINT CHR*(201): COLOR VGX
9110 REM ENTER? Ja, return
9120 IF I$ = CHR*(13) THEN RETURN
9130 REM Backspace (correctietoets)?
9140 IF I$ <> CHR*(8) GOTO 9230
9150 REM LI$ leeg? Niets wissen
9160 IF LEN(LI$) = 0 GOTO 9060
9170 REM Wis laatste karakter
9180 LI$ = LEFT*(LI$, LEN(LI$) - 1)
9190 REM Pas cursorpositie aan
9200 HOX = HOX - 6
9210 IF HOX < 0 THEN HOX = HOX + 252: VEX = VEX - 8
9220 GOTO 9060
9230 REM Juiste ASCII code? Nee, volgende toets
9240 IF I$ < CHR*(32) GOTO 9060
9250 REM Voeg I$ aan LI$ toe en pas de cursorpositie aan
9260 LOCATE HOX, VEX: PRINT I$: HOX = HOX + 6
9270 IF HOX > 251 THEN HOX = HOX - 252: VEX = VEX + 8
9280 LI$ = LI$ + I$: GOTO 9060
9290 ' -----

```

Deze subroutine is bijna het equivalent van de LINE INPUT instructie; er kan echter geen prompt string meegegeven worden (eerst PRINTen dus) en de beroemde SCREEN EDITOR werkt niet (de backspace toets wel).

Voor het aanroepen van deze subroutine moet SCREEN 1 actief zijn en kunnen de mooiste plaatjes op het scherm getoerd worden, waarna met GOSUB 9000 de gebruiker zijn gegevens kan intoetsen.

Waar op gelet moet worden is, dat deze subroutine een "cursor" gebruikt die wordt gewist met de achtergrondkleur van de laatste COLOR instructie. Als om de een of andere reden tijdens GOSUB 9000 deze cursor in een gebied komt waar de achtergrondkleur anders is (na LINE ..., BF of zo) dan ontstaat er een kleurfout op de plaats van de cursor.

De subroutine levert een string (LI\$), waarvan de waarde (als een getal ingevoerd is) met VAL(LI\$) opgevraagd kan worden. Als u meerdere gegevens tegelijk (net als bij INPUT) wil intoetsen, dient u nog een routine te maken die de diverse gegevens in de string uit elkaar haalt.

---

---

Ferdy Bossy

---

---

```
100 REM : worteltrekken met de macht van Uw keuze
110 REM : 2e machtwortel is SQR; maar
120 REM: de 10de machts wortel uit 328 kan natuurlijk ook.
130 CLS
140 PRINT"      ***** WORTEL TREKKEN *****"
150 PRINT
160 PRINT"      *****"
170 PRINT:PRINT:PRINT:PRINT
180 INPUT"      de wortel uit ";M:PRINT:PRINT
190 INPUT "de wortel met de macht ";W
200 PRINT:PRINT:PRINT:PRINT
210 PRINT"de";W"e machts wortel uit";M;"=":PRINT:PRINT USING"####.####";M^(1/W)
230 ' print using gebruikt omdat er anders vreemde uitkomsten komen.
240 PRINT:PRINT:INPUT "Nog een keer, tik ja of nee"; A$
250 IF A$="ja" THEN 130
255 IF A$="nee" THEN 270
260 IF A$=<>"ja" AND A$=<>"nee" THEN 240
270 PRINT:PRINT"Tot de volgende keer"
280 END
```

~~~~~SINTAKS ERROR~~~~~

Wat ging er fout in onze vorige uitgave? Dat was in de eerste plaats het programmaatje links midden in informatie bulletin nr. 3. We geven hier de juiste listing:

```
100 SCREEN 1
110 FOR X = 1 TO 15: LOCATE0,0: COLOR X
120 FOR Y = 32 TO 255: PRINT CHR*(Y);
130 NEXT Y, X
140 GOTO 140
150 END
```

Dat was niet dramatisch, maar het verschil valt wel op. Vervolgens bleek er op blz. 36 maar en halve sprite in te lezen te zijn. Dit kwam door regel 190 die moest zijn:

```
190 NEXT S: SPRITE*(I)= S* + T*
```

en dan gaat het wel goed.

Op bladzijde 35 had regel 170 moeten zijn:

```
170 IF MID$(A$, M, 1) > " "
    THEN C = C + 2 ^ (8-M)
```

En ook nog even iets over de computer himself. Het is velen al opgevallen dat PUT SPRITE niet overeenkomt met LOCATE, PSET etc.. De video processor zet nl. een sprite een positie lager dan opgegeven. Wanneer er een sprite gePUT dient te worden, dan zal het Y-coördinaat altijd met 1 verminderd dienen te worden, dus Y - 1 ten opzichte van het LOCATE coördinaat.

Vervolgens hebben we nog enige nuttige ROM- en start-adressen:

```
4198H karaktergenerator (tot 4550H)
02C9H tabel BASIC woorden (tot 0586H)
05D6H tabel foutmeldingen (tot 084BH)
0018H print karakter (accu) op beeldscherm
003EH karakter van toetsenbord lezen (in accu)
0044H print accu naar printer
0081H idem, print ev. extra spaties bij code 9
0047H SCREEN 0
004AH SCREEN 1
004DH SCREEN 2
0053H SCREEN ,0 (commando-lijn uit)
0056H SCREEN ,1 (commando-lijn aan)
005CH kijkt naar CTRL/STOP (carry 1: ingedrukt)
0147H beep
```


4782H Tekent SPECTRAVIDEO logo

Wie weet er nog meer?

```

10 CLS: INPUT "Hoeveel Punten": PNT
15 DIM TBL (PNT, 1): SCREEN 1
20 FOR TEL = 0 TO PNT
25   HLP = 6.28319 / PNT * TEL
30   TBL (TEL, 0) = 128 + 72 * SIN(HLP)
35   TBL (TEL, 1) = 96 + 95 * COS(HLP)
40   IF TEL = 0 THEN NEXT TEL
45   FOR HLP = 0 TO TEL
50     LINE (TBL (HLP, 0), TBL (HLP, 1))-
        (TBL (TEL, 0), TBL (TEL, 1))
55   NEXT HLP, TEL
60 GOTO 60

```



```

10 CLS: INPUT "Rantal Punten": PNT
20 SCREEN 1
30 FOR HOEK = -.2 TO 3.34 STEP 3.54 / PNT
40   H = 90 * COS (HOEK)
50   V = -60 * SIN (HOEK)
60   LINE (128, 80) -STEP (H, V)
70   LINE -STEP(0, 70): LINE -(128, 150)
80 NEXT HOEK
90 GOTO 90

```

